
!  Check whether a sports team can end the season with more
  wins than any other team, and thus win the pennant ;
! We maximize Z = excess wins of team Kchk over every other team.
  If we solve this model as an LP and Z <= 0, then clearly team Kchk
  cannot win the pennant outright;
! Ref: Robinson, L. (1991), "Baseball playoff eliminations: An application
  of linear programming," Operations Research Letters, North Holland, vol 10, pp. 67-74;
!  Wayne, K,(2001), "A New Property and a Faster Algorithm for Baseball Elimination,"
  SIAM J. Discrete Math., vol 14 no 2, pp 223-229;
! Keywords: Baseball, Elimination, LINGO, MLB, Pennant, Playoff, Robinson, Sports, Wayne;
SETS:
  TEAM: WINS, WFin;
  TxT( TEAM, TEAM) : NRem, W ;
ENDSETS

DATA:
 MAKINT  = 0 ; ! 1: wins must be an integer, 0: can be fractional;
! The teams;
!Case01  TEAM = CARDS  CUBS CINCY  METS;
! Wins so far, each team;
!Case01    WINS =  79     79    81    81;
!Case01    KChk = 1;  ! Check whether the CARDS can win most games;
! Games remaining for each i and j;
!Case01  
    NREM =  0      3     0     0
            3      0     0     0
            0      0     0     3
            0      0     3     0 ;
 ! The CARDS could finish  with 79 + 3 = 82 wins.
  This would beat CINCY if CINCY lost all remaining games.
  Could CARDS have a better record than every other team?;

! The teams;
!Case02;  TEAM = ATLANTA PHILLY  NEWYORK MONTREAL;
! Wins so far, each team;
!Case02;    WINS =  83     79    78    76;
!Case02;    KChk = 4;  ! Check whether the Montreal can win most games;
! Games remaining for each i and j. Matrix must be symmetric;
!Case02;  
    NREM =  0      1     6     1
            1      0     0     3
            6      0     0     1
            1      3     1     0 ;
ENDDATA

SUBMODEL StillPoss:
! Is it still possible for team KChk to win more games than any other
  team by the end of the regular season;
 ! Parameters:
      WINS( j) = wins so far by team j,
      LOSS( j) = losses so far by team j,
      NRem( i, j) = number remaining games between i and j,
      NTot = total games in season played by each team,
      KChk = index of the team we wish to check,
  Variables:
     W( i, j) = number times team i wins against team j
                in remaining games,
     WFin( j) = final number of wins for team j
      Z  = number games by which team KChk has more wins than any other team;

  @FOR( TxT( i, j) | i #LT# j:
! Each pair plays each other the requisite number remaining games;
    [PLAY]  W( i, j) + W( j, i) = NREM( i, j);
       );

  @FOR( TxT( i, j) | MAKINT #AND# i #LT# j:
         @GIN( W( i, j)); ! W should be a general integer;
       );



 ! Compute final wins for each team;
  @FOR( TEAM( i):
    [FINAL] WFin( i) = WINS( i) + @SUM( TxT( i, j)  | i #NE# j: W( i, j));
      );
  @FREE( Z); ! It could be negative;
! Z <= difference for every other team j;
! Difference in final wins of team KChk - team i;
  @FOR( TEAM( j) | j #NE# KChk:
    [BEAT] Z <= WFin( KChk) - WFin( j);
      );
   MAX = Z;   
ENDSUBMODEL

CALC:
  @SET( 'TERSEO',2); ! Output level (0:verb, 1:terse, 2:only errors, 3:none);
  @SET( 'PRBLVL', 1);! Integer probing level (0:LINGO, 1:none, 7:high);
!  @GEN( Stillposs);
  @SOLVE( StillPoss);
  @WRITE( ' Best possible excess wins for team ', TEAM( KChk), ' is ', Z, @NEWLINE( 1));
  @IFC( Z #LE# 0:
   @WRITE( TEAM ( KChk),' cannot win the pennant.', @NEWLINE( 1)); 
       );
ENDCALC


